<commit_before>cmake_minimum_required(VERSION 3.0.0)
project(grml VERSION 0.0.0)

## Compiler
set(CMAKE_CXX_STANDARD 17)
set(CMAKE_CXX_STANDARD_REQUIRED ON)
set(CMAKE_CXX_EXTENSIONS OFF)

## Project Dependencies
include(vendor/boost/CMakeLists.txt)

## Libraries
include_directories(lib)

# Applications
add_executable(${PROJECT_NAME} bin/grml.cpp)
target_link_libraries(${PROJECT_NAME} Boost::boost)

# Tests
include(CTest)
enable_testing()

add_executable(test_parser lib/test_parser.cpp)
target_link_libraries(test_parser
    Boost::unit_test_framework
)
add_test(NAME test_parser COMMAND test_parser)

add_executable(test_inferer lib/inferer.cpp lib/test_inferer.cpp)
target_link_libraries(test_inferer
    Boost::unit_test_framework
)
add_test(NAME test_inferer COMMAND test_inferer)<commit_msg>Split out parser and inferer libraries<commit_after>cmake_minimum_required(VERSION 3.0.0)
project(grml VERSION 0.0.0)

## Compiler
set(CMAKE_CXX_STANDARD 17)
set(CMAKE_CXX_STANDARD_REQUIRED ON)
set(CMAKE_CXX_EXTENSIONS OFF)

## Project Dependencies
include(vendor/boost/CMakeLists.txt)

## Libraries
add_library(parser INTERFACE)
target_include_directories(parser INTERFACE lib)

add_library(inferer lib/inferer.cpp)
target_include_directories(inferer PUBLIC lib)
target_link_libraries(inferer parser)

# Applications
add_executable(${PROJECT_NAME} bin/grml.cpp)
target_link_libraries(${PROJECT_NAME} parser Boost::boost)

# Tests
include(CTest)
enable_testing()

add_executable(test_parser lib/test_parser.cpp)
target_link_libraries(test_parser
    parser
    Boost::unit_test_framework
)
add_test(NAME test_parser COMMAND test_parser)

add_executable(test_inferer lib/test_inferer.cpp)
target_link_libraries(test_inferer
    inferer
    Boost::unit_test_framework
)
add_test(NAME test_inferer COMMAND test_inferer)
{'fraction_non_alphanumeric': 0.08370044052863436, 'fraction_numerical': 0.00881057268722467, 'mean_word_length': 8.875, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '1721734', 'n_tokens_mistral': 704, 'n_tokens_neox': 660, 'n_words': 101}